
1 [bookmark: _Toc122434485][bookmark: _Toc132373550]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk12 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com 	

2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc132373551]Corrections required
2. [bookmark: _Toc106975634][bookmark: _Toc109680267][bookmark: _Toc122623878][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc132373552]Change 1
	Function name
	f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig

	Reason for change
	1, according to R5w230004, time “T” for DAPS handover is set to 700ms. 
The current implementation gets the timing for T before calling f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig().
Somwhow for the TTCN execution, from the start of f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig, it cost about 150ms to reach step 0 (exclusive, ie. all inside TTCN, without ASP call to SS). 
As a result, T is effectively reduced to 550ms for all SS configurations

2, this function has parameter p_RBConfig_KeyChange , but it is not passed on to function call f_NR_GetPdcpCountInfoListIntraNR_HO at step 0

	Summary of change
	1, get T just before step 1 and return it for future usage
2, updated call f_NR_GetPdcpCountInfoListIntraNR_HO at step 0 with p_RBConfig_KeyChange , passed on

	TTCN module
	NR5GC_Handover

	MCC160 Comment
	



Before Change
	  function f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig(NR_CellId_Type p_SourceCellId,
                                                       NR_CellId_Type p_TargetCellId,
                                                       SubFrameTiming_Type p_Timing,  // timing based on Source Cell
                                                       boolean p_RBConfig_KeyChange := true) runs on NR5GC_PTC return NR_PdcpCountInfoList_Type
  {
    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;
    var template (value) NR_DRBInfoList_Type v_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();
    var template (value) NR_RadioBearerList_Type v_RadioBearerList := f_NR5GC_GetRadioBearerList(v_DRBInfoList);
    var template (value) DL_DCCH_Message v_RRCReconfiguration;
    var template (omit) MasterKeyUpdate v_MasterKeyUpdate := omit;
    var SubFrameTiming_Type v_Timing1;
    var SubFrameTiming_Type v_Timing2;
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList; //@sic R5-230116 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_SRBs; //@sic R5-230116 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_DRB; //@sic R5-230116 sic@
    var template(value) TimingInfo_Type v_TimingInfo; //@sic R5s230023 sic@
    
    v_Timing1 := f_SubFrameTiming_AddMilliSeconds (p_Timing, 5);  // time for sending RRCReconfiguration
    v_Timing2 := f_SubFrameTiming_AddMilliSeconds (p_Timing, 10); // time for COUNT tranfer and configuration in target cell
    
    //Acc. Table 8.1.4.3.1.3.3-1
    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(p_SourceCellId, tsc_NR_BWP_Id);
    // @sic R5s221151 sic@ Deleted "v_RACH_ConfigGeneric.preambleTransMax := n200;"
    if (p_RBConfig_KeyChange == true)  { v_MasterKeyUpdate := cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount); }
    v_RRCReconfiguration := f_NR5GC_RRCReconfiguration_IntraNR_HO_DAPS(p_TargetCellId,
                                                                       -,
                                                                       v_MasterKeyUpdate,
                                                                       cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))),
                                                                       -,
                                                                       p_RBConfig_KeyChange);
    // Step 0. Source Cell:     Get PDCP COUNT for the DAPS DRB and SRBs
    v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_SourceCellId, v_RadioBearerList);   //RBConfig_KeyChange:=true, TimingInfo_Now, IncrementSRB1:=true
    
    // Step 1. Target Cell: Configuration of DRBs
......... 
.........


}


[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc66864323][bookmark: _Toc84011981]
After Change
	  function f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig(inout SubFrameTiming_Type p_Timing, // timing based on Source Cell 
                                                                                                                                    NR_CellId_Type p_SourceCellId,
                                                       NR_CellId_Type p_TargetCellId,
                                                       boolean p_RBConfig_KeyChange := true) runs on NR5GC_PTC return NR_PdcpCountInfoList_Type
  {
    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;
    var template (value) NR_DRBInfoList_Type v_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();
    var template (value) NR_RadioBearerList_Type v_RadioBearerList := f_NR5GC_GetRadioBearerList(v_DRBInfoList);
    var template (value) DL_DCCH_Message v_RRCReconfiguration;
    var template (omit) MasterKeyUpdate v_MasterKeyUpdate := omit;
    var SubFrameTiming_Type v_Timing1;
    var SubFrameTiming_Type v_Timing2;
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList; //@sic R5-230116 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_SRBs; //@sic R5-230116 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_DRB; //@sic R5-230116 sic@
    var template(value) TimingInfo_Type v_TimingInfo; //@sic R5s230023 sic@
    /* MOVED to later
    v_Timing1 := f_SubFrameTiming_AddMilliSeconds (p_Timing, 5);  // time for sending RRCReconfiguration
v_Timing2 := f_SubFrameTiming_AddMilliSeconds (p_Timing, 10); // time for COUNT tranfer and configuration in target cell
*/
    
    //Acc. Table 8.1.4.3.1.3.3-1
    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(p_SourceCellId, tsc_NR_BWP_Id);
    // @sic R5s221151 sic@ Deleted "v_RACH_ConfigGeneric.preambleTransMax := n200;"
    if (p_RBConfig_KeyChange == true)  { v_MasterKeyUpdate := cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount); }
    v_RRCReconfiguration := f_NR5GC_RRCReconfiguration_IntraNR_HO_DAPS(p_TargetCellId,
                                                                       -,
                                                                       v_MasterKeyUpdate,
                                                                       cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))),
                                                                       -,
                                                                       p_RBConfig_KeyChange);
    // Step 0. Source Cell:     Get PDCP COUNT for the DAPS DRB and SRBs
    v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_SourceCellId, v_RadioBearerList, p_RBConfig_KeyChange);   //RBConfig_KeyChange:=true, TimingInfo_Now, IncrementSRB1:=true
    

    p_Timing := f_NR_GetNextSendOccasion_DL(p_SourceCellId, 700);
    v_Timing1 := f_SubFrameTiming_AddMilliSeconds (p_Timing, 5);  // time for sending RRCReconfiguration
v_Timing2 := f_SubFrameTiming_AddMilliSeconds (p_Timing, 10); // time for COUNT tranfer and configuration in target cell

    // Step 1. Target Cell: Configuration of DRBs

......... 
......... 


}



2. [bookmark: _Toc132373553]Change 2
	Function name
	f_NR5GC_DAPS_HO_CheckDataPath_HARQFeedback

	Reason for change
	Step 14B and 14C, the NR_SYSTEM_IND and NR_DRB_COMMON_IND may reach TTCN out of order due to race condition

	Summary of change
	Put NR_SYSTEM_IND and NR_DRB_COMMON_IND into interleave

	TTCN module
	NR5GC_Handover	

	MCC160 Comment
	



Before Change
	function f_NR5GC_DAPS_HO_CheckDataPath_HARQFeedback(NR_CellId_Type p_SourceCellId,
                                                      charstring p_StepNo := "",
                                                      NR_CellId_Type p_TargetCellId) runs on NR5GC_PTC //@sic R5s221151 sic@
  {
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var template (value) QosFlow_Identification_Type v_QosFlowId;
    var IntegerList_Type v_DrbList;
    var PDUSessionInfoList_Type v_PDUSessionInfoList;
    timer t_Watchdog := 5.0;

    v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_DrbList := f_NR_GetDRBList_ForGivenPDUSession(oct2int(v_PDUSessionInfoList[0].SessionId)); // Get list of DRBs for the first PDU session
    v_QosFlowId := f_NR_GetQosFlowIdentification_ForDRBId(v_DrbList[0]); //Get the QosFlow ID information from the DRB ID
   
    // Configure SS to report reception of HARQ ACK or NACK
    f_NR_SS_SystemIndCtrlConfig(p_SourceCellId, cds_NR_SystemIndCtrl_UL_HARQ(enable));
    
    // Step 14A. Source Cell: Send one IP packet on the DAPS DRB (first DRB of the first PDU session)
    DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(p_SourceCellId, v_QosFlowId, {v_IPData}));
    
    // Step 14B. Source Cell: Check that the UE transmit a positive HARQ feedback
    SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_SourceCellId, ack));
    f_NR_PreliminaryPass(__FILE__, __LINE__, p_StepNo);
    
    // Step 14C. Target Cell: Receive loopbacked IP packet
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(p_TargetCellId, v_QosFlowId, {v_IPData})); //@sic R5s221151 sic@
    t_Watchdog.stop;
  }



After Change
	  function f_NR5GC_DAPS_HO_CheckDataPath_HARQFeedback(NR_CellId_Type p_SourceCellId,
                                                      charstring p_StepNo := "",
                                                      NR_CellId_Type p_TargetCellId) runs on NR5GC_PTC //@sic R5s221151 sic@
  {
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var template (value) QosFlow_Identification_Type v_QosFlowId;
    var IntegerList_Type v_DrbList;
    var PDUSessionInfoList_Type v_PDUSessionInfoList;
    timer t_Watchdog := 5.0;

    v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_DrbList := f_NR_GetDRBList_ForGivenPDUSession(oct2int(v_PDUSessionInfoList[0].SessionId)); // Get list of DRBs for the first PDU session
    v_QosFlowId := f_NR_GetQosFlowIdentification_ForDRBId(v_DrbList[0]); //Get the QosFlow ID information from the DRB ID
   
    // Configure SS to report reception of HARQ ACK or NACK
    f_NR_SS_SystemIndCtrlConfig(p_SourceCellId, cds_NR_SystemIndCtrl_UL_HARQ(enable));
    
    // Step 14A. Source Cell: Send one IP packet on the DAPS DRB (first DRB of the first PDU session)
    DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(p_SourceCellId, v_QosFlowId, {v_IPData}));
    
    interleave {
        // Step 14B. Source Cell: Check that the UE transmit a positive HARQ feedback
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_SourceCellId, ack)) {
          f_NR_PreliminaryPass(__FILE__, __LINE__, p_StepNo);
        }
        // Step 14C. Target Cell: Receive loopbacked IP packet
        [] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(p_TargetCellId, v_QosFlowId, {v_IPData})) {} //@sic R5s221151 sic@
    }
    t_Watchdog.stop;
  }



2. [bookmark: _Toc132373554]Change 3
	Function name
	f_NR5GC_DAPS_HO_SourceRelease

	Reason for change
	With the current TTCN implementation, not only PDCP count of DRB, but also PDCP Count of SRBs are transferred at step 17. 

	Summary of change
	Correct to just transfer PDCP count of DRB

	TTCN module
	NR5GC_Handover

	MCC160 Comment
	



Before Change
	  function f_NR5GC_DAPS_HO_SourceRelease(NR_CellId_Type p_SourceCellId,
                                         NR_CellId_Type p_TargetCellId,
                                         charstring p_StepNo := "",
                                         boolean p_AdjustULcount := true) runs on NR_BASE_PTC
  { //@sic R5-230116 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;
    
    // Step 15. Target Cell: Send RRCReconfiguration (daps_SourceRelesase) to release the Source Cell)
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_38508_RRCReconfiguration_DAPS_HO_ReleaseSource));
    
    // Step 16. Target Cell: Receive RRCReconfigurationComplete)
    SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetCellId, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, p_StepNo);
    
    // Step 17. Transfer of the PDCP Count for DAPS DRB from source to target cell:
    //    a)    Source Cell: Get PDCP COUNT.
    //    b)    Target Cell: Set PDCP COUNT.
     v_NR_PdcpCountInfoList:=f_NR_SS_PdcpCount_Get(p_SourceCellId);
     
    //if Steps 14A-14C of procedure 7.3.5.3.5 took place, the UL count needs to be adjusted
    if (p_AdjustULcount) { //For DAPS DRB, the PDCP COUNT is maintained and incremented by one in UL to consider the data exchanged in steps 14A-14C.
      v_NR_PdcpCountInfoList[2].UL.Value := int2bit((bit2int(v_NR_PdcpCountInfoList[2].UL.Value) + 1), 32); //@sic R5s230025 sic@
    }
    f_NR_SS_PdcpCount_Set(p_TargetCellId, v_NR_PdcpCountInfoList, cs_TimingInfo_Now);
  }



After Change
	  function f_NR5GC_DAPS_HO_SourceRelease(NR_CellId_Type p_SourceCellId,
                                         NR_CellId_Type p_TargetCellId,
                                         charstring p_StepNo := "",
                                         boolean p_AdjustULcount := true) runs on NR_BASE_PTC
  { //@sic R5-230116 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_Source; 
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_Target; 
    
    // Step 15. Target Cell: Send RRCReconfiguration (daps_SourceRelesase) to release the Source Cell)
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_38508_RRCReconfiguration_DAPS_HO_ReleaseSource));
    
    // Step 16. Target Cell: Receive RRCReconfigurationComplete)
    SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetCellId, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, p_StepNo);
    
    // Step 17. Transfer of the PDCP Count for DAPS DRB from source to target cell:
    //    a)    Source Cell: Get PDCP COUNT.
    //    b)    Target Cell: Set PDCP COUNT.
    //KS R5xxx change_8143x
    v_NR_PdcpCountInfoList_Source:=f_NR_SS_PdcpCount_Get(p_SourceCellId);
    v_NR_PdcpCountInfoList_Target:=f_NR_SS_PdcpCount_Get(p_TargetCellId);
    
    v_NR_PdcpCountInfoList_Target[2].DL.Value := v_NR_PdcpCountInfoList_Source[2].DL.Value;
    v_NR_PdcpCountInfoList_Target[2].UL.Value := v_NR_PdcpCountInfoList_Source[2].UL.Value;
     
    //if Steps 14A-14C of procedure 7.3.5.3.5 took place, the UL count needs to be adjusted
    if (p_AdjustULcount) { //For DAPS DRB, the PDCP COUNT is maintained and incremented by one in UL to consider the data exchanged in steps 14A-14C.
      v_NR_PdcpCountInfoList_Target[2].UL.Value := int2bit((bit2int(v_NR_PdcpCountInfoList_Source[2].UL.Value) + 1), 32); //@sic R5s230025 sic@
    }
    f_NR_SS_PdcpCount_Set(p_TargetCellId, v_NR_PdcpCountInfoList_Target, cs_TimingInfo_Now);
  }	



2. [bookmark: _Toc132373555]Change 4
	Function name
	fl_TC_8_1_4_3_1And4_TestBody

	Reason for change
	1, same as 1.1
2, At step8. UE discards the PDCP PDU for DRB data, because the PDCP PDU is not ciphered

	Summary of change
	1, Get T in f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig

2, Active ciphering for DRB

	TTCN module
	RRC_DAPS_Handover_NR5GC

	MCC160 Comment
	




Before Change
	  function fl_TC_8_1_4_3_1And4_TestBody(NR_CellId_Type p_SourceCellId,
                                        NR_CellId_Type p_TargetCellId) runs on NR5GC_PTC
  { //This function follows the RRC procedure described in TS 38.523-3, 7.3.5.3.5
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing3;
    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(p_SourceCellId, tsc_NR_BWP_Id);
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_DRB;
    v_Timing := f_NR_GetNextSendOccasion_DL(p_SourceCellId, 700); // 700 ms
    v_Timing3 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 15); // time for sending IP packets to source cell

    //@siclog "Step 1" siclog@
    //The SS transmits an RRCReconfiguration message containing reconfigurationWithSync to order UE to perform
    //DAPS handover to NR Cell 2. DRB#n of the first PDU session is configured as DAPS bearer.
    v_NR_PdcpCountInfoList_DRB := f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig(p_SourceCellId, p_TargetCellId, v_Timing); //TS 38.523-3, 7.3.5.3.5, Steps 1-8, R5w230004: this function now returns v_NR_PdcpCountInfoList instead of v_RadioBearerList
    
    //@siclog "Step 2" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 [4] Table 4.9.1-1 indicate that the UE is capable
    //       of exchanging IP data on DRB#n of the first PDU session on NR Cell 1?
    f_NR5GC_DAPS_HO_CheckDataPath(p_SourceCellId, "Step 2", cs_TimingInfo_NR(v_Timing3)); //TS 38.523-3, 7.3.5.3.5, Steps 9-10
    //After data exchange, transfer count to the target cell
    //f_NR_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_RadioBearerList, -, cs_TimingInfo_Now, false); //@sic R5s221151 sic@
    v_NR_PdcpCountInfoList_DRB[0].DL.Value := int2bit((bit2int(v_NR_PdcpCountInfoList_DRB[0].DL.Value) + 1), 32);
    v_NR_PdcpCountInfoList_DRB[0].UL.Value := int2bit((bit2int(v_NR_PdcpCountInfoList_DRB[0].UL.Value) + 1), 32);
    f_NR_SS_PdcpCount_Set(p_TargetCellId, v_NR_PdcpCountInfoList_DRB, cs_TimingInfo_Now);  //Set count for DAPS DRB; TS 38.523-3, 7.3.5.3.5, Step 10A
 
    //@siclog "Steps 4-5" siclog@
    f_NR5GC_DAPS_HO_RACHProcedure_RRCReconfigurationComplete(p_TargetCellId,  //TS 38.523-3, 7.3.5.3.5, Steps 11-14
                                                             "Step 5",
                                                             -,
                                                             cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    //@siclog "Steps 5A-5B" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath_HARQFeedback(p_SourceCellId, "Step 5A", p_TargetCellId); //@sic R5s221151 sic@
    
    //@siclog "Steps 6-7" siclog@
    f_NR5GC_DAPS_HO_SourceRelease(p_SourceCellId, p_TargetCellId, "Step 7"); //TS 38.523-3, 7.3.5.3.5, Steps 15-17, @sic R5-230116 sic@

    //@siclog "Step 8" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath(p_TargetCellId, "Step 8"); //TS 38.523-3, 7.3.5.3.5, Step 18
    //@siclog "Steps 9-10" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath_NoReply(p_SourceCellId); //TS 38.523-3, 7.3.5.3.5, Step 18
    f_NR5GC_DAPS_HO_SourceRelease_SSConfig(p_SourceCellId); //TS 38.523-3, 7.3.5.3.5, Steps 19-20
  }



After Change
	  function fl_TC_8_1_4_3_1And4_TestBody(NR_CellId_Type p_SourceCellId,
                                        NR_CellId_Type p_TargetCellId) runs on NR5GC_PTC
  { //This function follows the RRC procedure described in TS 38.523-3, 7.3.5.3.5
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing3;
    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(p_SourceCellId, tsc_NR_BWP_Id);
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_DRB;
    var NR_SecurityParams_Type v_RRC_SecurityParams;

    //@siclog "Step 1" siclog@
    //The SS transmits an RRCReconfiguration message containing reconfigurationWithSync to order UE to perform
    //DAPS handover to NR Cell 2. DRB#n of the first PDU session is configured as DAPS bearer.
[bookmark: OLE_LINK11]    v_NR_PdcpCountInfoList_DRB := f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig(v_Timing, p_SourceCellId, p_TargetCellId); //TS 38.523-3, 7.3.5.3.5, Steps 1-8, R5w230004: this function now returns v_NR_PdcpCountInfoList instead of v_RadioBearerList
    v_Timing3 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 15); // time for sending IP packets to source cell   
     
    //@siclog "Step 2" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 [4] Table 4.9.1-1 indicate that the UE is capable
    //       of exchanging IP data on DRB#n of the first PDU session on NR Cell 1?
    f_NR5GC_DAPS_HO_CheckDataPath(p_SourceCellId, "Step 2", cs_TimingInfo_NR(v_Timing3)); //TS 38.523-3, 7.3.5.3.5, Steps 9-10
    //After data exchange, transfer count to the target cell
    //f_NR_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_RadioBearerList, -, cs_TimingInfo_Now, false); //@sic R5s221151 sic@
    v_NR_PdcpCountInfoList_DRB[0].DL.Value := int2bit((bit2int(v_NR_PdcpCountInfoList_DRB[0].DL.Value) + 1), 32);
    v_NR_PdcpCountInfoList_DRB[0].UL.Value := int2bit((bit2int(v_NR_PdcpCountInfoList_DRB[0].UL.Value) + 1), 32);
    f_NR_SS_PdcpCount_Set(p_TargetCellId, v_NR_PdcpCountInfoList_DRB, cs_TimingInfo_Now);  //Set count for DAPS DRB; TS 38.523-3, 7.3.5.3.5, Step 10A
 
    //@siclog "Steps 4-5" siclog@
    f_NR5GC_DAPS_HO_RACHProcedure_RRCReconfigurationComplete(p_TargetCellId,  //TS 38.523-3, 7.3.5.3.5, Steps 11-14
                                                             "Step 5",
                                                             -,
                                                             cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    //@siclog "Steps 5A-5B" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath_HARQFeedback(p_SourceCellId, "Step 5A", p_TargetCellId); //@sic R5s221151 sic@
    
    //@siclog "Steps 6-7" siclog@
    f_NR5GC_DAPS_HO_SourceRelease(p_SourceCellId, p_TargetCellId, "Step 7"); //TS 38.523-3, 7.3.5.3.5, Steps 15-17, @sic R5-230116 sic@

    v_RRC_SecurityParams:= f_NR_Security_Get();
    v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_TargetCellId, false);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_TargetCellId,
                                          v_RRC_SecurityParams.AS_Integrity,
                                          v_RRC_SecurityParams.AS_Ciphering);
    //@siclog "Step 8" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath(p_TargetCellId, "Step 8"); //TS 38.523-3, 7.3.5.3.5, Step 18
    //@siclog "Steps 9-10" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath_NoReply(p_SourceCellId); //TS 38.523-3, 7.3.5.3.5, Step 18
    f_NR5GC_DAPS_HO_SourceRelease_SSConfig(p_SourceCellId); //TS 38.523-3, 7.3.5.3.5, Steps 19-20
  }




